
8 Stochastic Methods

This chapter presents a variety of stochastic methods that use randomization strate-
gically to help explore the design space for an optimum. Randomness can help es-
cape local optima and increase the chances of finding a global optimum. Stochastic
methods typically use pseudo-random number generators to ensure repeatability.1 1 Although pseudo-random num-

ber generators produce numbers
that appear random, they are ac-
tually a result of a determinis-
tic process. Pseudo-random num-
bers can be produced through calls
to the rand function. The process
can be reset to an initial state us-
ing the seed! function from the
Random.jl package.

A large amount of randomness is generally ineffective because it prevents us from
effectively using previous evaluation points to help guide the search. This chapter
discusses a variety of ways to control the degree of randomness in our search.

8.1 Noisy Descent

Adding stochasticity to gradient descent can be beneficial in large nonlinear
optimization problems. Saddle points, where the gradient is very close to zero,
can cause descent methods to select step sizes that are too small to be useful. One
approach is to add Gaussian noise at each descent step:2 2 G. Hinton and S. Roweis,

“Stochastic Neighbor Embedding,”
in Advances in Neural Information
Processing Systems (NIPS), 2003.

x(k+1) ← x(k) + αg(k) + ǫ(k) (8.1)

where ǫ(k) is zero-mean Gaussian noise with standard deviation σ. The amount
of noise is typically reduced over time. The standard deviation of the noise is
typically a decreasing sequence σ(k) such as 1/k.3 Algorithm 8.1 provides an 3 The Hinton and Roweis paper

used a fixed standard deviation for
the first 3,500 iterations and set the
standard deviation to zero there-
after.

implementation of this method. Figure 8.1 compares descent with and without
noise on a saddle function.

A common approach for training neural networks is stochastic gradient descent,
which uses a noisy gradient approximation. In addition to helping traverse past
saddle points, evaluating noisy gradients using randomly chosen subsets of the
training data4 is significantly less expensive computationally than calculating the 4 These subsets are called batches.
true gradient at every iteration.

mutable struct NoisyDescent <: DescentMethod
submethod
σ
k

end
function init!(M::NoisyDescent, f, ∇f, x)

init!(M.submethod, f, ∇f, x)
M.k = 1
return M

end
function step!(M::NoisyDescent, f, ∇f, x)

x = step!(M.submethod, f, ∇f, x)
σ = M.σ(M.k)
x += σ.*randn(length(x))
M.k += 1
return x

end

Algorithm 8.1. A noisy descent
method, which augments another
descent method with additive
Gaussian noise. The method
takes another DescentMethod
submethod, a noise sequence σ,
and stores the iteration count k.

Convergence guarantees for stochastic gradient descent require that the positive
step sizes be chosen such that:

∞

∑
k=1

α(k) = ∞
∞

∑
k=1

(

α(k)
)2

< ∞ (8.2)

These conditions ensure that the step sizes decrease and allow the method to
converge, but not too quickly so as to become stuck away from a local minimum. x1

x
2

stochastic gradient descent
steepest descent

Figure 8.1. Adding stochasticity
to a descent method helps with
traversing saddle points such as
f (x) = x2

1 − x2
2 shown here. Due

to the initialization, the steepest
descent method converges to the
saddle point where the gradient is
zero.

8.2 Mesh Adaptive Direct Search

The generalized pattern search methods covered in section 7.4 restricted local
exploration to a fixed set of directions. In contrast, mesh adaptive direct search uses
random positive spanning directions.5

5 This section follows the lower
triangular mesh adaptive direct
search given by C. Audet and J. E.
Dennis Jr., “Mesh Adaptive Direct
Search Algorithms for Constrained
Optimization,” SIAM Journal onOp-
timization, vol. 17, no. 1, pp. 188–
217, 2006.

The procedure used to sample positive spanning sets (see example 8.1) begins
by constructing an initial linearly spanning set in the form of a lower triangular
matrix L. The diagonal terms in L are sampled from ±1/

√
α(k), where α(k) is the

step size at iteration k. The lower components of L are sampled from
{

−1/
√

α(k) + 1,−1/
√

α(k) + 2, . . . , 1/
√

α(k) − 1
}

(8.3)

Consider positive spanning sets constructed from the nonzero directions
d1, d2 ∈ {−1, 0, 1}. There are 8 positive spanning sets with 3 elements that
can be constructed from these directions:

Example 8.1. Positive spanning
sets for R2. Note that the lower
triangular generation strategy can
only generate the first two columns
of spanning sets.

The rows and columns of L are then randomly permuted to obtain a matrix D

whose columns correspond to n directions that linearly span Rn. The maximum
magnitude among these directions is 1/

√
α(k).

Two common methods for obtaining a positive spanning set from the linearly
spanning set are to add one additional direction d(n+1) = −∑

n
i=1 d(i) and to add

n additional directions d(n+j) = −d(j) for j in {1, . . . , n}. We use the first method
in algorithm 8.2.

The step size α starts at 1, is always a power of 4, and never exceeds 1. Using a
power of 4 causes the maximum possible step size taken in each iteration to be
scaled by a factor of 2, as the maximum step size α/

√
α has length 4m/

√
4m = 2m

for integer m < 1. The step size is updated according to:

α(k+1) ←







α(k)/4 if no improvement was found in this iteration
min(1, 4α(k)) otherwise

(8.4)
Mesh adaptive direct search is opportunistic but cannot support dynamic

ordering6 since, after a successful iteration, the step size is increased, and another 6 See section 7.4.
step in the successful direction would lie outside of the mesh. The algorithm
queries a new design point along the accepted descent direction. If f (x(k) =

x(k−1)+ αd) < f (x(k−1)), then the queried point is x(k−1) + 4αd = x(k) + 3αd. The

function rand_positive_spanning_set(α, n)
δ = round(Int, 1/sqrt(α))
L = Matrix(Diagonal(δ*rand([1,-1], n)))
for i in 1 : n-1

for j in 1:i-1
L[i,j] = rand(-δ+1:δ-1)

end
end
D = L[randperm(n),:]
D = D[:,randperm(n)]
D = hcat(D, -sum(D,dims=2))
return [D[:,i] for i in 1 : n+1]

end

Algorithm 8.2. Randomly sam-
pling a positive spanning set of
n + 1 directions according to mesh
adaptive direct search with step
size α and number of dimensions
n.

procedure is outlined in algorithm 8.3. Figure 8.2 illustrates how this algorithm
explores the search space.

8.3 Simulated Annealing

Simulated annealing7 borrows inspiration from metallurgy.8 Temperature is used 7 S. Kirkpatrick, C.D. Gelatt Jr.,
and M.P. Vecchi, “Optimization
by Simulated Annealing,” Science,
vol. 220, no. 4598, pp. 671–680,
1983.
8 Annealing is a process in which a
material is heated and then cooled,
making it more workable. When
hot, the atoms in the material are
more free to move around, and,
through random motion, tend to
settle into better positions. A slow
cooling brings the material to an
ordered, crystalline state. A fast,
abrupt quenching causes defects
because the material is forced to
settle in its current condition.

to control the degree of stochasticity during the randomized search. The tem-
perature starts high, allowing the process to freely move about the search space,
with the hope that in this phase the process will find a good region with the
best local minimum. The temperature is then slowly brought down, reducing
the stochasticity and forcing the search to converge to a minimum. Simulated
annealing is often used on functions with many local minima due to its ability to
escape local minima.

At every iteration, a candidate transition from x to x′ is sampled from a transi-
tion distribution T and is accepted with probability







1 if ∆y ≤ 0

e−∆y/t if ∆y > 0
(8.5)

where ∆y = f (x′)− f (x) is the difference in the objective and t is the temperature.
It is this acceptance probability, known as the Metropolis criterion, that allows the
algorithm to escape from local minima when the temperature is high.

function mesh_adaptive_direct_search(f, x, ϵ)
α, y, n = 1, f(x), length(x)
while α > ϵ

improved = false
for (i,d) in enumerate(rand_positive_spanning_set(α, n))

x′ = x + α*d
y′ = f(x′)
if y′ < y

x, y, improved = x′, y′, true
x′ = x + 3α*d
y′ = f(x′)
if y′ < y

x, y = x′, y′
end
break

end
end
α = improved ? min(4α, 1) : α/4

end
return x

end

Algorithm 8.3. Mesh adaptive di-
rect search for an objective function
f , an initial design x, and a toler-
ance ϵ.

Figure 8.2. Mesh adaptive direct
search proceeding left to right and
top to bottom.

Figure 8.3. Several annealing sched-
ules commonly used in simulated
annealing. The schedules have an
initial temperature of 10.

The temperature parameter t controls the acceptance probability. An annealing
schedule is used to slowly bring down the temperature as the algorithm pro-
gresses, as illustrated by figure 8.3. The temperature must be brought down to
ensure convergence. If it is brought down too quickly, the search method may not
cover the portion of the search space containing the global minimum.

It can be shown that a logarithmic annealing schedule of t(k) = t(1) ln(2)/ ln(k+ 1)

for the kth iteration is guaranteed to asymptotically reach the global optimum
under certain conditions,9 but it can be very slow in practice. The exponential

9 B. Hajek, “Cooling Schedules for
Optimal Annealing,” Mathematics
of Operations Research, vol. 13, no. 2,
pp. 311–329, 1988.

annealing schedule, which is more common, uses a simple decay factor:

t(k+1) = γt(k) (8.6)

for some γ ∈ (0, 1). Another common annealing schedule, fast annealing,10 uses a

10 H. Szu and R. Hartley, “Fast Sim-
ulated Annealing,” Physics Letters
A, vol. 122, no. 3-4, pp. 157–162,
1987.

temperature of
t(k) =

t(1)

k
(8.7)

A basic implementation of simulated annealing is provided by algorithm 8.4.
Example 8.2 shows the effect different transition distributions and annealing
schedules have on the optimization process.

function simulated_annealing(f, x, T, t, k_max)
y = f(x)
x_best, y_best = x, y
for k in 1 : k_max

x′ = x + rand(T)
y′ = f(x′)
Δy = y′ - y
if Δy ≤ 0 || rand() < exp(-Δy/t(k))

x, y = x′, y′
end
if y′ < y_best

x_best, y_best = x′, y′
end

end
return x_best

end

Algorithm 8.4. Simulated anneal-
ing, which takes as input an objec-
tive function f, an initial point x, a
transition distribution T, an anneal-
ing schedule t, and the number of
iterations k_max.

A more sophisticated algorithm was introduced by Corana et al. in 1987 that
allows for the step size to change during the search.11 Rather than using a fixed

11 A. Corana, M. Marchesi, C. Mar-
tini, and S. Ridella, “Minimizing
Multimodal Functions of Contin-
uous Variables with the ‘Simu-
latedAnnealing’Algorithm,”ACM
Transactions on Mathematical Soft-
ware, vol. 13, no. 3, pp. 262–280,
1987.

© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



8.3. s imulated annealing 131

We can use simulated annealing to optimize Ackley’s function, appendix B.1.
Ackley’s function has many local minima, making it easy for gradient-based
methods to get stuck.

Suppose we start at x(1) = [15, 15] and run 100 iterations. Below we show
the distribution over iterations for multiple runs with different combinations
of three zero-mean, diagonal covariance (σI) Gaussian transition distribu-
tions, and three different temperature schedules t(k) = t(1)/k.

In this case, the spread of the transition distribution has the greatest impact
on performance.

Example 8.2. Exploring the effect
of distribution variance and tem-
perature on the performance of
simulated annealing. The blue re-
gions indicate the 5% to 95% and
25% to 75% empirical Gaussian
quantiles of the objective function
value.

transition distribution, this adaptive simulated annealing method keeps track of
a separate step size v for each coordinate direction. For a given point x, a cycle of
random moves is performed in each coordinate direction i according to:

x′ = x + rviei (8.8)

where r is drawn uniformly at random from [−1, 1] and vi is the maximum step
size in the ith coordinate direction. Each new point is accepted according to the
Metropolis criterion. The number of accepted points in each coordinate direction
is stored in a vector a.

After ns cycles, the step sizes are adjusted with the aim to maintain an approxi-
mately equal number of accepted and rejected designs with an average acceptance
rate near one-half. Rejecting too many moves is a waste of computational effort,
while accepting too many moves indicates that the configuration is evolving too
slowly because candidate points are too similar to the current location. The update
formula used by Corana et al. is:

vi =







vi

(

1 + ci
ai/ns−0.6

0.4

)

if ai > 0.6ns

vi

(

1 + ci
0.4−ai/ns

0.4

)−1
if ai < 0.4ns

vi otherwise
(8.9)

The ci parameter controls the step variation along each direction and is typically
set to 2 as shown in figure 8.4. An implementation of the update is shown in
algorithm 8.5.

Figure 8.4. The step multiplication
factor as a function of acceptance
rate for c = 2.

function corana_update!(v, a, c, ns)
for i in 1 : length(v)

ai, ci = a[i], c[i]
if ai > 0.6ns

v[i] *= (1 + ci*(ai/ns - 0.6)/0.4)
elseif ai < 0.4ns

v[i] /= (1 + ci*(0.4-ai/ns)/0.4)
end

end
return v

end

Algorithm 8.5. The update for-
mula used by Corana et al. in adap-
tive simulated annealing, where v
is a vector of coordinate step sizes,
a is a vector of the number of ac-
cepted steps in each coordinate di-
rection, c is a vector of step scaling
factors for each coordinate direc-
tion, and ns is the number of cycles
before running the step size adjust-
ment.

Temperature reduction occurs every nt step adjustments, which is every ns · nt

cycles. The original implementation simply multiplies the temperature by a re-
duction factor.

The process is terminated when the temperature sinks low enough such that
improvement can no longer be expected. Termination occurs when themost recent
function value is no farther than ǫ from the previous nǫ iterations and the best
function value obtained over the course of execution. Algorithm 8.6 provides an
implementation and the algorithm is visualized in figure 8.5.

8.4 Cross-Entropy Method

The cross-entropy method,12 in contrast with the methods we have discussed so 12 R. Y. Rubinstein and D. P. Kroese,
The Cross-Entropy Method: A Unified
Approach to Combinatorial Optimiza-
tion, Monte-Carlo Simulation, and
Machine Learning. Springer, 2004.

far in this chapter, maintains an explicit probability distribution over the design
space.13 This probability distribution, often called a proposal distribution, is used to

13 The name of this method comes
from the fact that the process of
fitting the distribution involves
minimizing cross-entropy, which is
also called the Kullback–Leibler di-
vergence. Under certain conditions,
minimizing the cross-entropy cor-
responds to finding the maximum
likelihood estimate of the parame-
ters of the distribution.

propose new samples for the next iteration. At each iteration, we sample from the
proposal distribution and then update the proposal distribution to fit a collection
of the best samples. The aim at convergence is for the proposal distribution to
focus on the global optima. Algorithm 8.7 provides an implementation.

The cross-entropy method requires choosing a family of distributions parame-
terized byθ. One common choice is the family ofmultivariate normal distributions
parameterized by a mean vector and a covariance matrix. The algorithm also
requires us to specify the number of elite samples, melite, to use when fitting the
parameters for the next iteration.

Depending on the choice of distribution family, the process of fitting the distri-
bution to the elite samples can be done analytically. In the case of the multivariate
normal distribution, the parameters are updated according to the maximum
likelihood estimate:

µ(k+1) =
1

melite

melite
∑
i=1

x(i) (8.10)

Σ
(k+1) =

1

melite

melite
∑
i=1

(x(i) − µ(k+1))(x(i) − µ(k+1))⊤ (8.11)

Example 8.3 applies the cross-entropy method to a simple function. Figure 8.6
shows several iterations on a more complex function. Example 8.4 shows the
potential limitation of using a multivariate normal distribution for fitting elite
samples.

function adaptive_simulated_annealing(f, x, v, t, ϵ;
ns=20, nϵ=4, nt=max(100,5length(x)),
γ=0.85, c=fill(2,length(x)) )

y = f(x)
x_best, y_best = x, y
y_arr, n, U = [], length(x), Uniform(-1.0,1.0)
a,counts_cycles,counts_resets = zeros(n), 0, 0

while true
for i in 1:n

x′ = x + basis(i,n)*rand(U)*v[i]
y′ = f(x′)
Δy = y′ - y
if Δy < 0 || rand() < exp(-Δy/t)

x, y = x′, y′
a[i] += 1
if y′ < y_best; x_best, y_best = x′, y′; end

end
end

counts_cycles += 1
counts_cycles ≥ ns || continue

counts_cycles = 0
corana_update!(v, a, c, ns)
fill!(a, 0)
counts_resets += 1
counts_resets ≥ nt || continue

t *= γ
counts_resets = 0
push!(y_arr, y)

if !(length(y_arr) > nϵ && y_arr[end] - y_best ≤ ϵ &&
all(abs(y_arr[end]-y_arr[end-u]) ≤ ϵ for u in 1:nϵ))

x, y = x_best, y_best
else

break
end

end
return x_best

end

Algorithm 8.6. The adaptive simu-
lated annealing algorithm, where
f is the multivariate objective func-
tion, x is the starting point, v is
starting step vector, t is the start-
ing temperature, and ϵ is the termi-
nation criterion parameter. The op-
tional parameters are the number
of cycles before running the step
size adjustment ns, the number of
cycles before reducing the temper-
ature nt, the number of succes-
sive temperature reductions to test
for termination nϵ, the tempera-
ture reduction coefficient γ, and the
direction-wise varying criterion c.

Below is a flowchart for the
adaptive simulated annealing algo-
rithm as presented in the original
paper.

Initialize Parameters

Perform a cycle of random moves,
each along one coordinate direc-
tion. Accept or reject each point
according to the Metropolis crite-
rion. Record the optimum point
reached so far.

No. cycles ≥ ns?

Adjust step vector v.
Reset no. cycles to 0.

No. step adjustments ≥ nt?

Reduce temperature.
Reset no. adjustments to 0.
Add point to optimum list.

Stopping criterion satisfied?

End

yes

yes

yes

no

no

no

Figure 8.5. Simulated annealing
with an exponentially decaying
temperature, where the histograms
indicate the probability of simu-
lated annealing being at a partic-
ular position at that iteration.

using Distributions
function cross_entropy_method(f, P, k_max, m=100, m_elite=10)

for k in 1 : k_max
samples = rand(P, m)
order = sortperm([f(samples[:,i]) for i in 1:m])
P = fit(typeof(P), samples[:,order[1:m_elite]])

end
return P

end

Algorithm 8.7. The cross-entropy
method, which takes an objective
function f to be minimized, a pro-
posal distribution P, an iteration
count k_max, a sample size m, and
the number of samples to usewhen
refitting the distribution m_elite.
It returns the updated distribution
over where the global minimum is
likely to exist.

We can use Distributions.jl to represent, sample from, and fit proposal
distributions. The parameter vector θ is replaced by a distribution P. Calling
rand(P,m) will produce an n × m matrix corresponding to m samples of
n-dimensional samples from P, and calling fit will fit a new distribution of
the given input type.

import Random: seed!
import LinearAlgebra: norm
seed!(0) # set random seed for reproducible results
f = x->norm(x)
μ = [0.5, 1.5]
Σ = [1.0 0.2; 0.2 2.0]
P = MvNormal(μ, Σ)
k_max = 10
P = cross_entropy_method(f, P, k_max)
@show P.μ

P.μ = [-6.136226751273704e-7, -1.372160422208573e-6]

Example 8.3. An example of using
the cross-entropy method.

x1

x
2

x1 x1 x1

Figure 8.6. The cross-entropy
method with m = 40 applied to
the Branin function (appendix B.3)
using a multivariate Gaussian pro-
posal distribution. The 10 elite sam-
ples in each iteration are in red.© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.

2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



8.5. natural evolution strategies 137

The distribution family should be flexible enough to capture the relevant
features of the objective function. Here we show the limitations of using
a normal distribution on a multimodal objective function, which assigns
greater density in between the twominima. Amixture model is able to center
itself over each minimum.

Example 8.4. The normal distribu-
tion is unable to capture multiple
local minima, in contrast to mix-
ture models which can maintain
several.

8.5 Natural Evolution Strategies

Like the cross-entropy method, natural evolution strategies14 optimize a proposal 14 I. Rechenberg, Evolutionsstrate-
gie Optimierung technischer Systeme
nach Prinzipien der biologischen Evo-
lution. Frommann-Holzboog, 1973.

distribution parameterized by θ. We have to specify the proposal distribution
family and the number of samples. The aim is to minimize the expectation
Ex∼p(·|θ)[ f (x)]. Instead of fitting elite samples, evolution strategies apply gradient
descent. The gradient is estimated from the samples:15 15 This gradient estimation has re-

cently been successfully applied
to proposal distributions repre-
sented by deep neural networks.
T. Salimans, J. Ho, X. Chen, and
I. Sutskever, “Evolution Strate-
gies as a Scalable Alternative to
Reinforcement Learning,” ArXiv,
no. 1703.03864, 2017.

∇θ Ex∼p(·|θ)[ f (x)] =
∫

∇θp(x | θ) f (x) dx (8.12)

=
∫

p(x | θ)
p(x | θ)∇θp(x | θ) f (x) dx (8.13)

=
∫

p(x | θ)∇θ log p(x | θ) f (x) dx (8.14)

= Ex∼p(·|θ)[ f (x)∇θ log p(x | θ)] (8.15)

≈ 1

m

m

∑
i=1

f (x(i))∇θ log p(x(i) | θ) (8.16)

Although we do not need the gradient of the objective function, we do need the
gradient of the log likelihood, log p(x | θ). Example 8.5 shows how to compute

the gradient of the log likelihood for the multivariate normal distribution. The
estimated gradient can be used along with any of the descent methods discussed
in previous chapters to improve θ. Algorithm 8.8 uses gradient descent with a
fixed step size. Figure 8.7 shows a few iterations of the algorithm.

using Distributions
function natural_evolution_strategies(f, θ, k_max; m=100, α=0.01)

for k in 1 : k_max
samples = [rand(θ) for i in 1 : m]
θ -= α*sum(f(x)*∇logp(x, θ) for x in samples)/m

end
return θ

end

Algorithm 8.8. The natural evo-
lution strategies method, which
takes an objective function f to
be minimized, an initial distribu-
tion parameter vector θ, an itera-
tion count k_max, a sample size m,
and a step factor α. An optimized
parameter vector is returned. The
method rand(θ) should sample
from the distribution parameter-
ized by θ, and ∇logp(x, θ) should
return the log likelihood gradient.

8.6 Covariance Matrix Adaptation

Another popular method is covariance matrix adaptation,16 which is also referred to 16 It is common to use the phrase
evolution strategies to refer specifi-
cally to covariance matrix adapta-
tion.

as CMA-ES for covariance matrix adaptation evolutionary strategy. It has similarities
with natural evolution strategies from section 8.5, but the two should not be
confused. This method maintains a covariance matrix and is robust and sample
efficient. Like the cross-entropy method and natural evolution strategies, a distri-
bution is improved over time based on samples. Covariance matrix adaptation
uses multivariate Gaussian distributions.17 17 N. Hansen, “The CMA Evolu-

tion Strategy: A Tutorial,” ArXiv,
no. 1604.00772, 2016.Covariance matrix adaptation maintains a mean vector µ, a covariance matrix

Σ, and an additional step-size scalar σ. The covariance matrix only increases or
decreases in a single direction with every iteration, whereas the step-size scalar
is adapted to control the overall spread of the distribution. At every iteration, m

designs are sampled from the multivariate Gaussian:18 18 For optimization in Rn, it is rec-
ommended to use at least m =
4 + ⌊3 ln n⌋ samples per iteration,
and melite = ⌊m/2⌋ elite samples.

x ∼ N (µ, σ2
Σ) (8.17)

The designs are then sorted according to their objective function values such
that f (x(1)) ≤ f (x(2)) ≤ · · · ≤ f (x(m)). A new mean vector µ(k+1) is formed
using a weighted average of the first melite sampled designs:

µ(k+1) ←
melite
∑
i=1

wix
(i) (8.18)

The multivariate normal distribution N (µ, Σ) with mean µ and covariance
Σ is a popular distribution family due to having analytic solutions. The
likelihood in d dimensions has the form

p(x | µ, Σ) = (2π)−
d
2 |Σ|− 1

2 exp

(

−1

2
(x− µ)⊤Σ

−1(x− µ)

)

where |Σ| is the determinant of Σ. The log likelihood is

log p(x | µ, Σ) = −d

2
log(2π)− 1

2
log |Σ| − 1

2
(x− µ)⊤Σ

−1(x− µ)

The parameters can be updated using their log likelihood gradients:

∇(µ) log p(x | µ, Σ) = Σ
−1(x− µ)

∇(Σ) log p(x | µ, Σ) =
1

2
Σ
−1(x− µ)(x− µ)⊤Σ

−1 − 1

2
Σ
−1

The term ∇(Σ) contains the partial derivative of each entry of Σ with respect
to the log likelihood.

Directly updating Σ may not result in a positive definite matrix, as is re-
quired for covariance matrices. One solution is to represent Σ as a product
A⊤A, which guarantees that Σ remains positive semidefinite, and then up-
date A instead. Replacing Σ by A⊤A and taking the gradient with respect to
A yields:

∇(A) log p(x | µ, A) = A
[

∇(Σ) log p(x | µ, Σ) +∇(Σ) log p(x | µ, Σ)⊤
]

Example 8.5. A derivation of the
log likelihood gradient equations
for the multivariate Gaussian dis-
tribution. For the original deriva-
tion and severalmore sophisticated
solutions for handling the positive
definite covariance matrix, see D.
Wierstra, T. Schaul, T. Glasmachers,
Y. Sun, and J. Schmidhuber, “Nat-
ural Evolution Strategies,” ArXiv,
no. 1106.4487, 2011.

x1

x
2

x1 x1 x1

Figure 8.7. Natural evolution strate-
gies using multivariate Gaussian
distributions applied to Wheeler’s
Ridge, appendix B.7.

where the first melite weights sum to 1, and all the weights approximately sum to
0 and are ordered largest to smallest:19 19 In the recommended weighting,

the first melite samples are posi-
tive, and the remaining samples
are non-positive.

melite
∑
i=1

wi = 1
m

∑
i=1

wi ≈ 0 w1 ≥ w2 ≥ · · · ≥ wm (8.19)

Figure 8.8. Shown is an initial
proposal distribution (white con-
tours), six samples (white dots),
and the new updated means for
both covariance matrix adaptation
(blue dot) and the cross-entropy
method (red dot) using three elite
samples. Covariance matrix adap-
tation tends to update the mean
more aggressively than the cross-
entropy method (red dot), as it as-
signs higher weight to better sam-
pled designs, and negative weight
to worse sampled designs.

We can approximate the mean update in the cross-entropy method by setting
the first melite weights to 1/melite, and setting the remaining weights to zero. Co-
variance matrix adaptation instead distributes decreasing weight to all m designs,
including some negative weights. The recommended weighting is obtained by
normalizing

w′i = ln
m + 1

2
− ln i for i ∈ {1, . . . , m} (8.20)

to obtain w. The positive and negative weights are normalized separately. Fig-
ure 8.8 compares the mean updates for covariance matrix adaptation and the
cross-entropy method.

The step size is updated using a cumulative variable pσ that tracks steps over
time:

p
(1)
σ = 0

p
(k+1)
σ ← (1− cσ)pσ +

√

cσ(2− cσ)µeff(Σ
(k))−1/2δw

(8.21)

where cσ < 1 controls the rate of decay and the right hand term determines
whether the step size should be increased or decreased based on the observed
samples with respect to the present scale of the distribution. The variance effective
selection mass µeff has the form

µeff =
1

∑
melite
i=1 w2

i

(8.22)

and δw is computed from the sampled deviations:

δw =
melite
∑
i=1

wiδ
(i) for δ(i) =

x(i) − µ(k)

σ(k)
(8.23)

The new step size is obtained according to

σ(k+1) ← σ(k) exp

(
cσ

dσ

( ‖pσ‖
E‖N (0, I)‖ − 1

))

(8.24)

where

E‖N (0, I)‖ =
√

2
Γ
(

n+1
2

)

Γ
(

n
2

) ≈
√

n

(

1− 1

4n
+

1

21n2

)

(8.25)

is the expected length of a vector drawn from a Gaussian distribution. Comparing
the length of pσ to its expected length under random selection provides the
mechanism by which σ is increased or decreased. The constants cσ and dσ have
recommended values:

cσ = (µeff + 2)/(n + µeff + 5)

dσ = 1 + 2 max

(

0,
√

(µeff − 1)/(n + 1)− 1

)

+ cσ

(8.26)

The covariance matrix is also updated using a cumulative vector:

p
(1)
Σ = 0

p
(k+1)
Σ ← (1− cΣ)p

(k)
Σ + hσ

√

cΣ(2− cΣ)µeffδw

(8.27)

where

hσ =







1 if ‖pσ‖
√

1−(1−cσ)2(k+1)
<
(
1.4 + 2

n+1

)
E‖N (0, I)‖

0 otherwise
(8.28)

The hσ stalls the update of pΣ if ‖pΣ‖ is too large, thereby preventing excessive
increases in Σ when the step size is too small.

The update requires the adjusted weights w◦:

w◦i =







wi if wi ≥ 0
nwi

∥
∥
∥Σ−1/2δ(i)

∥
∥
∥

2 otherwise (8.29)

The covariance update is then

Σ
(k+1) ←




1 + c1cΣ(1− hσ)(2− cΣ)− c1 − cµ

︸ ︷︷ ︸

typically zero




Σ

(k) + c1pΣp⊤Σ
︸ ︷︷ ︸

rank-one update

+ cµ

µ

∑
i=1

w◦i δ
(i)
(

δ(i)
)⊤

︸ ︷︷ ︸

rank-µ update

(8.30)

The constants cΣ, c1 and cµ have recommended values

cΣ =
4 + µeff/n

n + 4 + 2µeff/n

c1 =
2

(n + 1.3)2 + µeff

cµ = min

(

1− c1, 2
µeff − 2 + 1/µeff
(n + 2)2 + µeff

)

(8.31)

The covariance update consists of three components: the previous covariance
matrix Σ

(k), a rank-one update, and a rank-µ update. The rank-one update gets
its name from the fact that pΣp⊤Σ has rank one; it has only one eigenvector along
pΣ. Rank-one updates using the cumulation vector allow for correlations between
consecutive steps to be exploited, permitting the covariance matrix to elongate
itself more quickly along a favorable axis.

The rank-µ update gets its name from the fact that ∑
µ
i=1 w◦i δ

(i)
(

δ(i)
)⊤

has rank
min(µ, n). One important difference between the empirical covariance matrix
update used by the cross-entropy method and the rank-µ update is that the
former estimates the covariance about the new mean µ(k+1), whereas the latter
estimates the covariance about the original mean µ(k). The δ(i) values thus help
estimate the variances of the sampled steps rather than the variance within the
sampled designs.

Covariance matrix adaptation is depicted in figure 8.9.

8.7 Summary

• Stochastic methods employ random numbers during the optimization process.

• Simulated annealing uses a temperature that controls random exploration and
which is reduced over time to converge on a local minimum.

• The cross-entropy method and evolution strategies maintain proposal distri-
butions from which they sample in order to inform updates.

• Natural evolution strategies uses gradient descent with respect to the log
likelihood to update its proposal distribution.

• Covariance matrix adaptation is a robust and sample-efficient optimizer that
maintains a multivariate Gaussian proposal distribution with a full covariance
matrix.

8.8 Exercises

Exercise 8.1. We have shown that mixture proposal distributions can better
capture multiple minima. Why might their use in the cross-entropy method be
limited?

function covariance_matrix_adaptation(f, x, k_max;
σ = 1.0,
m = 4 + floor(Int, 3*log(length(x))),
m_elite = div(m,2))

μ, n = copy(x), length(x)
ws = log((m+1)/2) .- log.(1:m)
ws[1:m_elite] ./= sum(ws[1:m_elite])
μ_eff = 1 / sum(ws[1:m_elite].^2)
cσ = (μ_eff + 2)/(n + μ_eff + 5)
dσ = 1 + 2max(0, sqrt((μ_eff-1)/(n+1))-1) + cσ
cΣ = (4 + μ_eff/n)/(n + 4 + 2μ_eff/n)
c1 = 2/((n+1.3)^2 + μ_eff)
cμ = min(1-c1, 2*(μ_eff-2+1/μ_eff)/((n+2)^2 + μ_eff))
ws[m_elite+1:end] .*= -(1 + c1/cμ)/sum(ws[m_elite+1:end])
E = n^0.5*(1-1/(4n)+1/(21*n^2))
pσ, pΣ, Σ = zeros(n), zeros(n), Matrix(1.0I, n, n)
for k in 1 : k_max

P = MvNormal(μ, σ^2*Σ)
xs = [rand(P) for i in 1 : m]
ys = [f(x) for x in xs]
is = sortperm(ys) # best to worst

# selection and mean update
δs = [(x - μ)/σ for x in xs]
δw = sum(ws[i]*δs[is[i]] for i in 1 : m_elite)
μ += σ*δw

# step-size control
C = Σ^-0.5
pσ = (1-cσ)*pσ + sqrt(cσ*(2-cσ)*μ_eff)*C*δw
σ *= exp(cσ/dσ * (norm(pσ)/E - 1))

# covariance adaptation
hσ = Int(norm(pσ)/sqrt(1-(1-cσ)^(2k)) < (1.4+2/(n+1))*E)
pΣ = (1-cΣ)*pΣ + hσ*sqrt(cΣ*(2-cΣ)*μ_eff)*δw
w0 = [ws[i]≥0 ? ws[i] : n*ws[i]/norm(C*δs[is[i]])^2

for i in 1:m]
Σ = (1-c1-cμ) * Σ +

c1*(pΣ*pΣ' + (1-hσ) * cΣ*(2-cΣ) * Σ) +
cμ*sum(w0[i]*δs[is[i]]*δs[is[i]]' for i in 1 : m)

Σ = triu(Σ)+triu(Σ,1)' # enforce symmetry
end
return μ

end

Algorithm 8.9. Covariance matrix
adaptation, which takes an objec-
tive function f to be minimized, an
initial design point x, and an itera-
tion count k_max. One can option-
ally specify the step-size scalar σ,
the sample size m, and the number
of elite samples m_elite.

The best candidate design point
is returned, which is the mean of
the final sample distribution.

The covariance matrix under-
goes an additional operation to en-
sure that it remains symmetric; oth-
erwise small numerical inconsis-
tencies can cause the matrix no
longer to be positive definite.

This implementation uses a
simplified normalization strategy
for the negative weights. The orig-
inal can be found in Equations
50–53 of N. Hansen, “The CMA
Evolution Strategy: A Tutorial,”
ArXiv, no. 1604.00772, 2016.

Figure 8.9. Covariancematrix adap-
tation using multivariate Gaussian
distributions applied to the flower
function, appendix B.4.

© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



8.8. exercises 145

Exercise 8.2. In the cross-entropy method, what is a potential effect of using an
elite sample size that is very close to the total sample size?

Exercise 8.3. The log-likelihood of a value sampled from a Gaussian distribution
with mean µ and variance ν is:

ℓ(x | µ, ν) = −1

2
ln 2π − 1

2
ln ν− (x− µ)2

2ν

Show why evolution strategies using Gaussian distributions may encounter diffi-
culties while applying a descent update on the variance when the mean is on the
optimum, µ = x∗.

Exercise 8.4. Derive the maximum likelihood estimate for the cross-entropy
method using multivariate normal distributions:

µ(k+1) =
1

m

m

∑
i=1

x(i)

Σ
(k+1) =

1

m

m

∑
i=1

(x(i) − µ(k+1))(x(i) − µ(k+1))⊤

where the maximum likelihood estimates are the parameter values that maximize
the likelihood of sampling the individuals

{

x(1), · · · , x(m)
}

.

